Math 467 Homework Set 1: some solutions

```
> with(DEtools): with(plots):
Warning, the name changecoords has been redefined
```

Problem 1: 2.2.7

Find the fixed points, determine their stability, for

$$\frac{d}{dt}\mathrm{x}(t) = \cos x - \mathbf{e}^x$$

```
> plot(cos(x) - exp(x),x = -5*Pi..Pi,y=-2..2);
```



It is easier to visualize plotting the curves of $\cos x$ and e^x separately:

```
> plot({cos(x),exp(x)},x = -5*Pi..Pi,y = -2..2);
```



The fixed points are at the intersections of these two curves. Let's find the first few numerically, specifying the intervals for the numerical solution, which we know from the behaviour of cos x:

```
> x0  := fsolve(exp(x)-cos(x),x,-1..1);
  x1  := fsolve(exp(x)-cos(x),x,-Pi..-1);
  x2  := fsolve(exp(x)-cos(x),x,-2*Pi..-Pi);
  x3  := fsolve(exp(x)-cos(x),x,-3*Pi..-2*Pi);
  x4  := fsolve(exp(x)-cos(x),x,-4*Pi..-3*Pi);
```

$$x0 := 0.$$

$$x1 := -1.292695719$$

$$x2 := -4.721292759$$

$$x3 := -7.853593280$$

$$x4 := -10.99559106$$

We observe that the roots xk for $2 \le k$ are close to the zeros of cos x, since e^x is small:

```
> evalf(seq(-(k-1/2)*Pi,k=2..4));
```

$$-4.712388981, -7.853981635, -10.99557429$$

From the sign of $e^x - \cos x$, we can see that the fixed points xk are stable if k is odd, and unstable if k is even.

We can deduce the qualitative behaviour of the solutions from the fixed points and their stability.

Direct numerical integration using Maple gives:

```
> eqp1 := diff(x(t),t) = exp(x(t)) - cos(x(t));
  initconds :=
  [[x(0)=1],[x(0)=0],[x(0)=-1],[x(0)=-2],[x(0)=-4],[x(0)=x2],[x(0
  )=-5],[x(0)=-7],
  [x(0)=-8]]:
  DEplot(eqp1,x(t),t=0..8,x=-9..2,initconds,linecolor=black);
>
```

$$eqp1 := \frac{d}{dt}\,x(t) = \mathbf{e}^{x(t)} - \cos(x(t))$$



A closed-form analytical solution is not available, since it would require the integration of $\dfrac{1}{e^x - \cos x}$.

Problem 2: 2.2.8

We seek a dynamical system yielding the given flow. A possible answer is given by

$$\frac{d\,x(t)}{dt} = (x+1)^2\,x(x-2)$$

```
> eqp2 := diff(x(t),t) = (x(t)+1)^2 * x(t) * (x(t)-2):
  initconds2 :=
  [[x(0)=-1.5],[x(0)=-1],[x(0)=-0.8],[x(0)=0],[x(0)=1.2],
  [x(0)=2],[x(0)=2.001]]:
  DEplot(eqp2,x(t),t=0..5,x=-2..3,initconds2,linecolor=black);
```



This is one of many possible answers; others are obtained by multiplying the vector field by $g(x)$, where g is positive except possibly at one or more of the fixed points.

Problem 3: 2.2.10 - Fixed Points

We seek examples of first-order dynamical systems `dx/dt = f(x)` satisfying certain conditions:

(a) Every real number is a fixed point:

$$\frac{dx}{dt} = 0$$

(b) Every integer is a fixed point, and there are no others:

$$\frac{dx}{dt} = \sin(\pi x)$$

(c) There are no examples in which there are exactly three fixed points, and all are stable. In fact, we cannot even have two stable fixed points adjacent to each other. This is easily seen by drawing a picture; but we can also argue as follows: Suppose x_1 and x_2 are two adjacent fixed points. Then, from the assumptions that f(x) is continuous, and that f(x) is nonzero between x_1 and x_2 (there are no fixed points between x_1 and x_2), we must have that f(x) has one sign between x_1 and x_2. Suppose without loss of generality that f(x)>0 in this interval. Then for any initial condition x_0 in this interval, the solution must be increasing (dx/dt > 0), and must approach x_2 as $t \to \infty$, consistent with the assumption that x_2 is stable. However, this contradicts the assumption that x_1 is stable, since we can choose the initial point x_0 as close to (but above) x_1 as we like.

(d) An equation with no fixed points is

$$\frac{dx}{dt} = 1$$

(e) A simple example of an equation with precisely 100 fixed points is

$$\frac{dx}{dt} = (x-1)(x-2) .. (x-99)(x-100)$$

Problem 4: 2.2.13 - Skydiver

```
> m := 'm': g := 'g': k := 'k':
  eqp3 := diff(v(t),t) = g - k*v(t)^2/m;
```

$$eqp3 := \frac{d}{dt} v(t) = g - \frac{k\, v(t)^2}{m}$$

The command odeadvisor indicates the solution method for this first-order ODE.

```
> odeadvisor(eqp3);
```

$$[\_quadrature]$$

To find the solution, we separate variables, and integrate; the integration of $(a^2 - v^2)^{(-1)}$(where $a = \sqrt{\frac{gm}{k}}$) is best performed using partial fractions. Maple can also solve this equation analytically:

```
dsolve(eqp3);
```

$$v(t) = \frac{\tanh\left(\frac{\sqrt{g\, m\, k}\,(t + \_C1)}{m}\right)\sqrt{g\, m\, k}}{k}$$

In fact, this solution is only valid if m, g and k are all positive, and if $-\sqrt{\frac{mg}{k}} < v(t)$. It seems that Maple automatically made these assumptions; in this case they are justified, but this example shows that in general Maple's analytical solutions are not always reliable: do the calculations by hand, and show your working! (you can use Maple to check, if you like). That is, the solution produced by Maple is only the general solution for $a < v(0)$ (the problem is to take care with absolute value signs ...). We can find the particular solution satisfying the initial condition $v(0) = 0$:

```
> solp3 := rhs(dsolve({eqp3,v(0)=0}));
```

$$solp3 := \frac{\tanh\left(\frac{\sqrt{g\, m\, k}\; t}{m}\right)\sqrt{g\, m\, k}}{k}$$

Now we can use Maple to find the asymptotic behaviour:

```
> limit(solp3,t=infinity);
```

$$\lim_{t \to \infty} \frac{\tanh\left(\frac{\sqrt{g\, m\, k}\; t}{m}\right)\sqrt{g\, m\, k}}{k}$$

Evidently, now (finally!) Maple is concerned about the sign of the variables. Let's try specifying that all variables are positive:

```
> limit(solp3,t=infinity) assuming (g>0,m>0,k>0);
```

$$\frac{\sqrt{g\,m\,k}}{k}$$

This gives the correct terminal velocity. We can write the formula for v(t) in terms of the terminal velocity V:

```
vsol := simplify(subs(m=k*V^2/g,solp3)) assuming (k>0,V>0);
```

$$vsol := \tanh\left(\frac{t\,g}{V}\right)V$$

This answer is much more easily obtained by the graphical method. We plot $\frac{dv}{dt}$ against v (choosing some values of the variables):

```
g := 10: m := 0.1: k := 1:
plot(g - k*v*v/m,v=-1.5..1.5);
m := 'm': g := 'g': k := 'k':
```



Now let's use the numbers given:
The average velocity is (in ft/sec)

```
Vavg := (31400-2100)/116; evalf(Vavg);
```

$$Vavg := \frac{7325}{29}$$

$$252.5862069$$

The distance travelled as a function of time is s(t) satisfying $\frac{ds}{dt} = v$ and s(0) = 0.

```
s1 := int(vsol,t) assuming (V>0,g>0);
```

$$s1 := -\frac{1}{2}\,\frac{V^2 \ln\left(\tanh\left(\frac{t\,g}{V}\right)-1\right)}{g} - \frac{1}{2}\,\frac{V^2 \ln\left(\tanh\left(\frac{t\,g}{V}\right)+1\right)}{g}$$

This solution doesn't look too correct: it is giving negative arguments of the ln function (and is thus complex-valued). Let's try to help Maple a bit, by performing the appropriate substitution by hand...

```
s2 := Int(vsol,t) assuming (V>0,g>0);

> s3 := value(student[changevar](tau=t*g/V,s2,tau));
> s := subs(tau=t*g/V,s3);
```

$$s2 := \int \tanh\left(\frac{t\,g}{V}\right) V\,dt$$

$$s3 := \frac{V^2 \ln(\cosh(\tau))}{g}$$

$$s := \frac{V^2 \ln\left(\cosh\left(\frac{t\,g}{V}\right)\right)}{g}$$

```
> g := 32.2: t := 116: s; dist := 31400-2100;
  solve(s=dist,V);
```

$$0.03105590062\, V^2 \ln\left(\cosh\left(\frac{3735.2}{V}\right)\right)$$

$$dist := 29300$$

$$-252.5862069$$

In the command solve, Maple attempts an analytical solution; in this case it gets it wrong (I'm not sure why; but the given value is the average velocity computed previously, which cannot also be the terminal velocity). For a problem with purely floating-point solutions, we should use fsolve (and look for the positive solution):

```
Vterm := fsolve(s=dist,V,V=0..infinity);
```

$$Vterm := 265.6854815$$

From this value of the terminal velocity, we can compute the drag constant *k*. Note that the weight (in pounds) is *mg*.

```
weight := 261.2:
kdrag := solve(sqrt(weight/k)=Vterm,k);
```

$$kdrag := 0.003700305037$$

```
> m := 'm': g := 'g': k := 'k': s := 's': t := 't':
```

Problem 5: 2.3.2 - Autocatalysis

The fixed points are readily found to be 0 and $\frac{k\_1\, a}{\mathrm{k\_}(-1)}$

```
> fp4 := k_1*a*x - km_1*x^2;
  solve(fp4,x);
```

$$fp4 := k\_1\, a\, x - km\_1\, x^2$$

$$0, \frac{k\_1\ a}{km\_1}$$

$x = 0$ is unstable, the other fixed point is stable.

We can do a quick graphical analysis, and plot some typical solutions, if we assume values for the constants:

```
a := 1: k_1 := 1: km_1 := 1:
plot(fp4,x=-0.5..1.5);
```




```
> DEplot(diff(x(t),t)=a*k_1*x(t)-km_1*x(t)^2,x(t),t=0..5,x=-0.2..
  2,[[x(0)=-0.0],[x(0)=0.1],[x(0)=0.7],[x(0)=1.7]],linecolor=blac
  k);
```



Problem 6: 2.4.7 - Pitchfork bifurcation

```
> f6 := (x,a) -> a*x - x^3;
```

$$f6 := (x, a) \rightarrow a\,x - x^3$$

We plot the three vector fields next to each other, using the array function:

```
p6a := plot(f6(x,-1),x=-1.5..1.5,y=-2..2,tickmarks=[0,0]):
p6b := plot(f6(x,0),x=-1.5..1.5,y=-2..2,tickmarks=[0,0]):
p6c := plot(f6(x,1),x=-1.5..1.5,y=-2..2,tickmarks=[0,0]):
```

```
> plots6 := array(1..1,1..3):
  plots6[1,1]:=p6a: plots6[1,2]:=p6b: plots6[1,3]:=p6c:
  display(plots6);
```



For $a \le 0$, there is a unique fixed point at $x = 0$, which is stable; for $a < 0$ this is found by linear stability analysis (since f'(0) = $a < 0$), while for $a = 0$, linear stability analysis does not prove stability (decay towards the origin is slower than exponential - see the next problem), but a look at the plot of $-x^3$ shows that the origin is stable.

If $0 < a$, there are three fixed points, at $x = 0$, $x = -\sqrt{a}$ and $x = \sqrt{a}$. Now f'(0) = a is positive, so the origin is unstable, while the other two fixed points are stable, with f' = $-2\,a$. This is also apparent from the graphs.

```
>
```

Problem 7: 2.4.9 - Critical slowing down

Reset variables:

```
x0 := 'x0':
```

```
> eqp7 := diff(x(t),t) = - x(t)^3;
```

$$eqp7 := \frac{d}{dt}\,\mathrm{x}(t) = -\mathrm{x}(t)^3$$

Find the analytical solution with arbitrary initial condition:

```
xsa := dsolve({eqp7,x(0)=x0},x(t)) assuming x0>0;
xsb := dsolve({eqp7,x(0)=x0},x(t)) assuming x0<0;
xsz := dsolve({eqp7,x(0)=0},x(t));
```

$$xsa := x(t) = \frac{1}{\sqrt{2\,t + \frac{1}{x0^2}}}$$

$$xsb := x(t) = -\frac{1}{\sqrt{2\,t + \frac{1}{x0^2}}}$$

$$xsz := x(t) = 0$$

```
> limit(xsa,t=infinity);
  limit(xsb,t=infinity);
```

$$\lim_{t \to \infty} x(t) = 0$$

$$\lim_{t \to \infty} x(t) = 0$$

So the solutions approach zero for arbitrary initial conditions; however, the decay is proportional to $\frac{1}{\sqrt{t}}$, not exponential.

We plot the solutions of this equation and of $\frac{dx}{dt} = -x$ on the same graph:

```
lineq := diff(x(t),t) = -x(t):
linsoln := dsolve({lineq,x(0)=10},x(t));
critsoln := dsolve({eqp7,x(0)=10},x(t));
```

$$linsoln := x(t) = 10\,\mathbf{e}^{(-t)}$$

$$critsoln := x(t) = \frac{1}{\sqrt{2\,t + \frac{1}{100}}}$$

```
> plot([rhs(linsoln),rhs(critsoln)],t=0..10);
```



Note that the solution to $\dfrac{dx}{dt} = -x^3$ decays much more rapidly initially, but then slows down once $x < 1$.

Problem 8: 2.5.1 - Reaching origin in finite time

The origin $x = 0$ is a stable fixed point for any real $0 < c$. We plot a few representative vector fields:

```
> plot(-x^(1/2),x=0..2,tickmarks=[0,0]);
  plot(-x^1,x=0..4,-4..0.5,tickmarks=[0,0]);
  plot(-x^2,x=0..2,y=-4..0.5,tickmarks=[0,0]);
```







We know that for $c = 1$, the decay towards the origin is exponential, and x approaches 0 asymptotically. When $1 < c$, the decay is slower than exponential, as we derived in Problem 7. So the only possibility for the solution to decay to zero in finite time is for $c < 1$.
The time taken from $x = 1$ to $x = 0$ is

```
T = int(-1/x^c,x=1..0);
```

$$T = -\left(\lim_{x \to 0+} -\frac{x^{(-c+1)} - 1}{c - 1}\right)$$

When $1 < c$, the limit diverges; when $c = 1$, T is also infinite ($T = -\mathit{lim}\ \ln x$). When $c < 1$, the time is finite:

```
T = int(-1/x^c,x=1..0) assuming c < 1;
```

$$T = -\frac{1}{c-1}$$

Problem 9: 2.5.2 - Blow-up

We know that solutions y(t) of $\frac{dy}{dt} = 1 + y^2$ blow up in finite time. Now for $1 < x$, the solutions x(t) of $\frac{dx}{dt} = 1 + x^{10}$ grow more rapidly than y(t), since $x^2 < x^{10}$ for $1 < x$. Thus the solutions x(t) must also blow up in finite time. This is not yet a complete argument, though, since it is only valid for $1 < x$; but since $1 \le 1 + x^{10}$, we know that solutions beginning at any initial condition $x0$ will reach $x = 1$ at the latest at time $t = 1 - x0$; and since we reach $x = 1$ in finite time, we can then begin the comparison with y(t).

An alternative argument: suppose x(0) = $x0$. The time taken to diverge (reach $x = \infty$) is given by

```
T = Int(1/(1+x^10),x=x0..infinity);
```

$$T = \int_{x0}^{\infty} \frac{1}{1+x^{10}}\,dx$$

If this is finite for all $x0$, then we have finite-time blow-up. But we have

```
T < Int(1/(1+x^10),x=-infinity..infinity):
```
so
```
Int(1/(1+x^10),x=-1..1) + 2*Int(1/(1+x^10),x=1..infinity):
```

and introducing appropriate comparisons, we find

```
T < Int(1/1,x=-1..1) + 2*Int(1/(1+x^2),x=1..infinity);
```

$$T < \int_{-1}^{1} 1\,dx + 2\int_{1}^{\infty} \frac{1}{1+x^2}\,dx$$

Thus an estimate of the upper bound for the blow-up time for any initial condition is

```
> int(1,x=-1..1) + 2*int(1/(1+x^2),x=1..infinity);   evalf(%);
```

(clearly finite)

$$2 + \frac{\pi}{2}$$

$$3.570796327$$

The actual upper bound is

```
int(1/(1+x^10),x=-infinity..infinity); evalf(%);
```

$$\frac{1}{5}\,\frac{\pi}{\sin\left(\frac{\pi}{10}\right)}$$

2.033281478

We plot some numerical solutions:

```
DEplot(diff(x(t),t)=1+x(t)^(10),x(t),t=0..3,x=-3..5,[[x(0)=-1.1]]
,stepsize=0.01,linecolor=black);
```



>